MAX3634 I/O Model

SPICE I/O Macromodels aid in understanding signal integrity issues in electronic systems. Most of Maxim's High Frequency/Fiber Communication ICs utilize input and output (I/O) circuits with Current Mode Logic (CML), Positive Emitter Coupled Logic (PECL), and Low Voltage Differential Signal (LVDS) formats to transfer data into and out of an IC. These models are based on simplified circuit expressions that may include replacement of active circuit elements with ideal controlled voltage and current sources. As such, simulation with macromodels should be treated as 'typical' performance and
not relied upon as final proof-of-design. Use of macromodel descriptions is not a substitute for worst-case design analysis, nor for testing real circuits over temperature, supply, and other operating limits.

The output format is provided as ASCII text netlists suitable for generic SPICE. This format is compatible with most versions of SPICE such as PSPICE and HSPICE. Additional information is found in HFAN 6.1 *Input/Output Models for Maxim Fiber Component*s.

To extract the circuit netlists using the Adobe Acrobat Reader follow these instructions. Select the "Text Select Tool" by clicking the left mouse button on this icon of the menu bar (a capital T with a small dashed box to the lower right). Highlight the desired netlist text with the cursor. Use the copy command from the edit menu to capture the selected lines. Then paste the selected lines into a text file editor and save the file with an extension compatible with the simulator.

Version A, Jan 27, 2005



Figure 1. Output signal buffer for the MAX3634 including a simplified package model.



Figure 2. Simplified input package model and input circuitry for the MAX3634.

Notes:

MAX3634 Input Model

This input model is suitable for monitoring the input signals SD1+, SD1-, REFCLK+ and REFCLK-.

The input pins are nodes 1000 (IN+) and 1001 (IN-). These are connected to a model pattern generator that includes]50Ω output impedances. The nodes 8 and 9 can be used to monitor the signal integrity.

MAX3634 PECL Output Model

The PECL output model is a simplified version of the output stage used by the MAX3634 limiting amplifier. The package model is quite accurate, and the output impedance presented by the output emitter-followers should also be fairly accurate. Simplifications have been made regarding performance over temperature, so the model is only accurate at a temperature of 35°C.

The output pins are nodes 2001 (OUT-) and 2002 (OUT+). In order to see the signal at the output a PECL load of 50Ω to (VCC – 2V) has been included. This is the external termination and not included on the MAX3634. Replace this load with the interconnect model for signal integrity simulations.

The enclosed output model is suitable for the data outputs SD0+, SD0- and clock outputs SCLK+ and SCLK-.

Text File Format: This model is shipped in "pdf" format. Models and netlists can be copied to text format in the Acrobat Reader by holding the left mouse button on the "Text Select Tool." Then the user can "select" what netlist and/or subcircuit with the mouse. Then use the copy command from the "edit" menu to capture the selected lines. These lines can then be "pasted" into the user's text file. Make sure to save the text tile as plain text without any formatting.

Circuit Netlist – Input circuit

```
INPUT - MAX3634 INPUT CIRCUIT
*
* This example is the input circuit for the MAX3634
*
.OPT ACCT NOMOD NOPAGE RELTOL=.001 LIMPTS=1001

.WIDTH OUT=80

* TEMP = 27 + 290mW/18.9 = 42
.TEMP 42
.OP
.TRAN 100PS 3NS
VCC 101 0 DC 3.3V

VINP 2 0 PULSE (1.3V 1.7V 0.3N 80P 80P 643P 1606P)
VINN 3 0 PULSE (1.7V 1.3V 0.3N 80P 80P 643P 1606P)


RLD1 2 1000 50
RLD2 3 1001 50

XPK1 8 9 1000 1001 INPKG

XCIROUT 10 11 8 9 101 INPUTA

.SUBCKT INPUTA 10 11 3001 3002 101

* The power supply is 5 Volts.
RB1 101 20 15K
RB2 0 20 21k
RB3 20 3001 4.7K
RB4 20 3002 4.7K
RB5 3001 3003 254
RB6 3002 3004 254

XQ1 10 3003 4 0 N102M024S
XQ2 11 3004 4 0 N102M024S
*N102M024 is a X2 device
RC1 10 101 203
RC2 11 101 203
IE1 4 0 1M

XQ3 3003 3003 300 0 N102M010
XQ4 300 300 3002 0 N102M010
XQ5 3004 3004 310 0 N102M010
XQ6 310 310 3003 0 N102M010

XPAD1 3001 0 101 0 PADESD100
XPAD2 3002 0 101 0 PADESD100

.ENDS INPUTA

.SUBCKT INPKG 201 202 101 102
*
* resistors
*
RB01 201 301 0.25
RB02 202 302 0.25
```

```
*
* inductors
*
LLAP_1_3 101 301 1.2N
LLAP_2_4 102 302 1.3N
K02_03 LLAP_1_3 LLAP_2_4 0.294

*LB03 PADT PADBOT 23P
*
* capacitors
*
C01 101 0 143F
C02 102 0 143F
*
* mutual capacitors
*
C01_02 101 102 43.800F
.ENDS INPKG

*
.SUBCKT N102M024S 1 2 3 21
CP1SUB 2 201 2.600F
RP1SUB 20 201 100K
CTRENCH 1 202 11.453F
RFIELDEPI 202 21 837.055
RREVERT 202 20 1G
CBL 10 20 2.024F
RSUB 20 21 252.365K
CWAFER 20 21 2.087F
CP1EPI 10 12 2.101F
CP1P2 12 3 2.100F
RBX 2 12 65.018 TC=2.271M
RCX 1 10 23.542 TC=2.717M,449.424N
RCI 10 11 5.885 TC=2.717M,449.424N
REX 13 3 17.198
QN 11 12 13 11 TX 2
.MODEL TX NPN( IS=2.558E-018 XTI=3 EG=1.120 BF=380 BR=12 XTB=0 VAF=66
+   VAR=2.500 NF=1.018 NR=1.020 NE=2 NC=1.560 IKF=5.628M IKR=159.900U
+   ISE=1.279E-018 ISC=0 RB=32.509 RBM=24.382 IRB=575.640U CJE=6.016F
+   MJE=463M VJE=1.100 FC=990M CJC=3.276F MJC=350M VJC=1 TF=1.320P TR=5N
+   XTF=2 VTF=1.200 ITF=20.787M PTF=5 KF=227.190N AF=2.150 )
.ENDS N102M024S
*
.SUBCKT N102M010 1 2 3 21
CP1SUB 2 201 1.212F
RP1SUB 20 201 100K
CTRENCH 1 202 5.008F
RFIELDEPI 202 21 1.913K
RREVERT 202 20 1G
CBL 10 20 6.901E-016
RSUB 20 21 619.308K
CWAFER 20 21 8.506E-016
CP1EPI 10 12 7.614E-016
CP1P2 12 3 6.631E-016
RBX 2 12 234.971 TC=2.423M
RCX 1 10 86.810 TC=2.815M,503.797N
RCI 10 11 21.703 TC=2.815M,503.797N
REX 13 3 79.826
QN 11 12 13 11 TX

.MODEL TX NPN( IS=1.102E-018 XTI=3 EG=1.120 BF=380 BR=12 XTB=0 VAF=66
```

```
+   VAR=2.500 NF=1.018 NR=1.020 NE=2 NC=1.560 IKF=2.425M IKR=68.900U
+   ISE=5.512E-019 ISC=0 RB=58.743 RBM=44.057 IRB=248.040U CJE=2.619F
+   MJE=463M VJE=1.100 FC=990M CJC=1.747F MJC=350M VJC=1 TF=1.320P TR=5N
+   XTF=2 VTF=1.200 ITF=8.957M PTF=5 KF=1.210U AF=2.150 )
.ENDS N102M010


.SUBCKT PADESD100 2 3 4 5
XP1 2 5 PAD4SQ3P7
XQ1 2 3 5 DE0900
XQ2 4 2 5 DE0900
.ENDS PADESD100

.SUBCKT PAD4SQ3P7 1 3
CPAD 1 10 67.534F
REPI 10 20 378.507 TC=4.800M,5U
CTRENCH 21 20 22.531F
CBL 21 20 1.413P
RX 20 21 1G
RS 3 21 5.416K
CWAFER 21 3 2.587F
.ENDS PAD4SQ3P7

.SUBCKT DE0900 1 2 21
CTRENCH 2 202 58.872F
RFIELDEPI 202 21 163.328
RREVERT 202 20 1G
CBL 4 5 78.946F
RSUB 5 21 26.725K
CWAFER 5 21 19.712F
CP1EPI 1 4 76.086F
DD 1 4 DCB
RS 4 2 5.368 TC=4.277M,4.217U
.MODEL DCB D( IS=7.920E-018 N=1.050 CJO=216F VJ=800M M=500M )
.ENDS DE0900

.PRINT TRAN  V(1001) V(1002)
.PROBE
.END
```

Circuit Netlist – Output circuit

```
INPUT -  MAX3634ETM OUTPUT CIRCUIT
*
* THIS IS THE TYPICAL PECL OUTPUT OF THE MAX3968CEP with package.
*
.OPTIONS ACCT NOMOD NOPAGE LIMPTS=10000 RELTOL=.001
.WIDTH OUT=80
.TEMP 34
.OP
.TRAN 100PS 2nS
*
* CONVENTIONS VCC = 101, VEE = 102, + OUT = 2000, - OUT = 2001
*
VCC 101 0 DC 5.0
VEE 102 0 DC 3.0

RLOAD1 2001 102 50
CLOAD1 2001 101 0.20P
```

```
CLOAD2 2002 101 0.20P
RLOAD2 2002 102 50
CLOAD3 2001 2002 0.05P
EOUT 80 0 2001 2002 1

XPK1 2001 2002 2010 2011 0 0 0 OUTPKG
XCIROUT 2010 2011 101 OUTDRV
*
.SUBCKT OUTDRV 71 72 101
VINP 2 0 PULSE (4.70 3.90 0.3N 80P 80P 643P 1606P)
VINN 3 0 PULSE (3.90 4.70 0.3N 80P 80P 643P 1606P)
*
RB1 2 22 50
CB1 22 0 .2p
* Adjusted to match waveform of data sheet
RB2 3 32 50
CB2 32 0 .2p
* Adjusted to match waveform of data sheet
*
XQ1 101 22 64 0 N402V066
XQ2 101 32 66 0 N402V066
*
*IE1 64 0 20M
*IE2 66 0 20M

RE1 64 71 .01
RE2 66 72 .01

*.SUBCKT PADESD100 2 3 4 5

XPAD1 71 0 101 0 PADESD100
XPAD2 72 0 101 0 PADESD100

.ENDS OUTDRV
*

.SUBCKT OUTPKG 101 102 201 202 401 402 403
*
* resistors
*
RB01 201 301 0.25
RB02 202 302 0.25
*
* inductors
*
LLAP_1_3 101 301 1.2N
LLAP_2_4 102 302 1.3N
K02_03 LLAP_1_3 LLAP_2_4 0.294

*LB03 PADT PADBOT 23P
*
* capacitors
*
C01 101 403 143F
C02 102 403 143F
*
* mutual capacitors
*
C01_02 101 102 43.800F
.ENDS OUTPKG
```

```
** BEGINNING OF PROCESS LIB
*
.SUBCKT PADESD100 2 3 4 5
XP1 2 5 PAD4SQ3P7
XQ1 2 3 5 DE0900
XQ2 4 2 5 DE0900
.ENDS PADESD100

.SUBCKT PAD4SQ3P7 1 3
CPAD 1 10 67.534F
REPI 10 20 378.507 TC=4.800M,5U
CTRENCH 21 20 22.531F
CBL 21 20 1.413P
RX 20 21 1G
RS 3 21 5.416K
CWAFER 21 3 2.587F
.ENDS PAD4SQ3P7
*
.SUBCKT DE0900 1 2 21
CTRENCH 2 202 58.872F
RFIELDEPI 202 21 163.328
RREVERT 202 20 1G
CBL 4 5 78.946F
RSUB 5 21 26.725K
CWAFER 5 21 19.712F
CP1EPI 1 4 76.086F
DD 1 4 DCB
RS 4 2 5.368 TC=4.277M,4.217U
.MODEL DCB D( IS=7.920E-018 N=1.050 CJO=216F VJ=800M M=500M )
.ENDS DE0900
*
*
.SUBCKT N402V066 1 2 3 21
*Scaled by 14X
CP1SUB 2 201 70.009F
RP1SUB 20 201 100K
CTRENCH 1 202 190.774F
RFIELDEPI 202 21 50.349
RREVERT 202 20 1G
CBL 10 20 98.193F
RSUB 20 21 11.393K
CWAFER 20 21 46.239F
CP1EPI 10 12 134.894F
CP1P2 12 3 119.243F
RBX 2 12 1.455 TC=1.681M
RCX 1 10 640.716M TC=2.493M,1.163U
RCI 10 11 160.179M TC=2.493M,1.163U
REX 13 3 226.875M
QN 11 12 13 11 TX 14
.MODEL TX NPN( IS=2.771E-017 XTI=3 EG=1.120 BF=380 BR=12 XTB=0 VAF=66
+   VAR=2.500 NF=1.018 NR=1.020 NE=2 NC=1.560 IKF=60.952M IKR=1.732M
+   ISE=1.385E-017 ISC=6.234E-030 RB=5.092 RBM=3.819 IRB=6.234M
+   CJE=64.826F MJE=463M VJE=1.100 FC=990M CJC=31.450F MJC=350M VJC=1
+   TF=1.320P TR=5N XTF=2 VTF=1.200 ITF=225.108M PTF=5 KF=1.790N
+   AF=2.150 )
.ENDS N402V066

*
.PRINT TRAN V(2001) V(2002)
```

```
*.PROBE
*
.END
```